
EXCEL FORMULAS

Note: The first two pages of this document contain testable material. Computer Science 1 students are responsible for knowing the testable material.                        - Major Hector Salas
What is a formula?

A formula is an equation that performs operations on worksheet data. Formulas can perform mathematical operations, such as addition and multiplication, or they can compare worksheet values or join text. Formulas can refer to other cells on the same worksheet, cells on other sheets in the same workbook, or cells on sheets in other workbooks. The following example adds 25 to the value in cell B4 and then divides the result by the sum of the values in cells D5, E5, and F5.

Constructing a formula

Formulas calculate values in a specific order. A formula in Microsoft Excel always begins with an equal sign (=). The equal sign tells Excel that the succeeding characters constitute a formula. Following the equal sign are the elements to be calculated (the operands), which are separated by calculation operators. Excel calculates the formula from left to right, according to a specific order for each operator in the formula. You can change the order of operations by using parentheses.

In the example below, the parentheses around the first part of the formula force Excel to calculate B4+25 first and then divide the result by the sum of the values in cells D5, E5, and F5.

=(B4+25)/SUM(D5:F5)

Referring to Cells

A formula can refer to constant values and to other cells. The cell that contains the formula is known as a dependent cell when its value depends on the values in other cells. For example, cell B2 is a dependent cell if it contains the formula =C2.

Whenever a cell that the formula refers to changes, the dependent cell also changes, by default. For example, if a value in any of the following cells changes, the result of the formula =B2+C2+D2 also changes.

If you use constant values in the formula instead of references to the cells (for example, =30+70+110), the result changes only if you modify the formula yourself.

Formulas can also refer to ranges of cells or to names or labels that represent cells or ranges. 

Using Worksheet Functions

Microsoft Excel contains many predefined, or built-in, formulas, which are known as functions. Functions can be used to perform simple or complex calculations. The most frequently used function is the SUM function, which is used to add the numbers in a range of cells.

The SUM worksheet function adds the values in the range D5:F5.

About constructing formulas

The structure or order of the elements in a formula determine the final result of the calculation. Formulas in Microsoft Excel follow a specific syntax, or order, that includes an equal sign (=) followed by the elements to be calculated (the operands), which are separated by calculation operators. Each operand can be a value that does not change (a constant value), a cell or range reference, a label, a name, or a worksheet function.

Excel performs the operations from left to right — according to the order of operator precedence — starting with the equal sign (=). You can control the order of calculation by using parentheses to group operations that should be performed first. For example, the following formula produces 11 because Excel calculates multiplication before addition. The formula multiplies 2 by 3 and then adds 5 to the result.

=5+2*3

In contrast, if you use parentheses to change the syntax, Excel adds 5 and 2 together and then multiplies the result by 3 to produce 21.

=(5+2)*3

The order in which Microsoft Excel performs operations in formulas

If you combine several operators in a single formula, Microsoft Excel performs the operations in the order shown in the following table. If a formula contains operators with the same precedence — for example, if a formula contains both a multiplication and division operator — Excel evaluates the operators from left to right. To change the order of evaluation, enclose the part of the formula to be calculated first in parentheses. For more information about calculation operators, click.

	Operator
	Description

	: (colon) 

  (single space)

, (comma)
	Reference operators

	–
	Negation (as in –1)

	%
	Percent

	^
	Exponentiation

	* and /
	Multiplication and division

	+ and – 
	Addition and subtraction

	&
	Connects two strings of text (concatenation)

	= < > <= >= <>
	Comparison


Calculation operators in formulas

Operators specify the type of calculation that you want to perform on the elements of a formula. Microsoft Excel includes four different types of calculation operators: arithmetic, comparison, text, and reference.

Arithmetic operators   To perform basic mathematical operations such as addition, subtraction, or multiplication; combine numbers; and produce numeric results, use the following arithmetic operators.

	Arithmetic
operator
	
Meaning
	
Example

	+ (plus sign)
	Addition
	3+3

	– (minus sign)
	Subtraction
Negation
	3–1
–1

	* (asterisk)
	Multiplication
	3*3

	/ (forward slash)
	Division
	3/3

	% (percent sign)
	Percent
	20%

	^ (caret)
	Exponentiation
	3^2 (the same as 3*3)


Comparison operators   You can compare two values with the following operators. When two values are compared by using these operators, the result is a logical value, either TRUE or FALSE.

	Comparison
operator
	
Meaning
	
Example

	= (equal sign)
	Equal to
	A1=B1

	> (greater than sign)
	Greater than
	A1>B1

	< (less than sign)
	Less than
	A1<B1

	>= (greater than or equal to sign)
	Greater than or equal to
	A1>=B1

	<= (less than or equal to sign)
	Less than or equal to
	A1<=B1

	<> (not equal to sign)
	Not equal to
	A1<>B1


Text concatenation operator   Use the ampersand (&) to join, or concatenate, one or more text strings to produce a single piece of text.

	Text
operator
	
Meaning
	
Example

	& (ampersand)
	Connects, or concatenates, two values to produce one continuous text value
	"North" & "wind" produce "Northwind"


Reference operators   Combine ranges of cells for calculations with the following operators.

	Reference
operator
	
Meaning
	
Example

	: (colon)
	Range operator, which produces one reference to all the cells between two references, including the two references
	B5:B15

	, (comma)
	Union operator, which combines multiple references into one reference
	SUM(B5:B15,D5:D15)



Troubleshoot formulas and error values

If a formula cannot properly evaluate a result, Microsoft Excel will display an error value. For example, error values can be the result of using text where a formula expects a numeric value, deleting a cell that is referenced by a formula, or using a cell that is not wide enough to display the result.

Error values might not be caused by the formula itself. For example, if a formula shows #N/A or #VALUE!, a cell referenced by the formula may contain the error. You can locate cells that provide error values to other formulas by using the auditing tools.

What does the error ##### mean?
A ##### error value occurs when the cell contains a number, date, or time that is wider than the cell or when the cell contains a date and/or time formula that produces a negative result.

Increase the width of the column   You can resize the column by dragging the boundary between the column headings.

Apply a different number format   In some cases, you can change the cell's number format to make the number fit within the existing cell width. For example, decrease the number of decimal places after the decimal point. 

What does the error #VALUE! mean?
The #VALUE! error value occurs when the wrong type of argument or operand is used, or if the Formula AutoCorrect feature cannot correct the formula.

What does the error #NAME? mean?
The #NAME? error value occurs when Microsoft Excel doesn't recognize text in a formula.

Examples of commonly used formulas

Calculate Pay and Taxes

Hours Worked *  Pay Rate = Gross Pay

Gross Pay  * Tax Rate = Taxes

Gross Pay - SUM(Authorized Deductions) = Net Pay

Taxes may include social society taxes, medicare, and federal withholding taxes. All of these are considered authorized deductions that your employer will deduct from your gross pay. What you take home after taxes is called “net pay”. 
Calculate the running balance

You can create a checkbook register in Excel to track your banking transactions. As part of the spreadsheet, you can build a formula to calculate your running balance. In this example, assume that cell F6 contains the previous balance, cell D7 contains the first transaction's deposit subtotal, and cell E7 contains any cash-received amount.

To calculate the current balance for the first transaction, enter the following formula in cell F7:

=SUM(F6,D7,–E7)

As you enter new transactions, copy this formula into the current balance cell for each new transaction.

Join first and last names

You can join a list of first names in one column with a list of last names in another column with a formula. In these examples, assume that cell D5 contains the first name and cell E5 contains the last name.

To display the full name in the format "first_name last_name" (for example, "John Smith"):

=D5&" "&E5

To display the full name in the format "last_name, first_name" (for example, "Smith, John"):

=E5&", "&D5

Join a date with text

To join, or concatenate, two values to produce one continuous text value, use the & (ampersand) text operator. To connect a number, date, or time value stored in a cell with a text string, use the TEXT worksheet function. For example, if cell F5 contains a billing date of 5-Jun-96, you can display the text "Statement date: 5-Jun-96" by using this formula:

="Statement date: "&TEXT(F5, "d-mmm-yy")

Increase a number by a percentage

You can increase a numeric value stored in a cell by a percentage, such as 5 percent. In this example, assume that cell F5 contains the original value.

=F5*(1+5%)

If the percentage amount is stored in a cell (for example, cell F2):

=F5*(1+$F$2)

The reference to cell F2 is an absolute cell reference so that the formula can be copied to other cells without changing the reference to F2.

Create a total based on one condition

Use the SUMIF worksheet function to create a total value for one range based on a value in another range. For example, for every cell in the range B5:B25 that contains the value "Northwind", the following formula calculates the total for the corresponding cells in the range F5:F25.

=SUMIF(B5:B25,"Northwind",F5:F25)

Create a total based on multiple conditions

Use the following formula to calculate the total value of cells F5:F25, where B5:B25 contains "Northwind" and the range C5:C25 contains the region name "Western".

=SUM(IF((B5:B25="Northwind")*(C5:C25="Western"),F5:F25))

To calculate the total value of cells F5:F25, where B5:B25 contains either "Northwind" or "Terra Firm", use the following formula.

=SUM(IF((B5:B25="Northwind")+(B5:B25="Terra Firm"),F5:F25))

Both of these formulas are array formulas and must be entered by pressing CTRL+SHIFT+ENTER. Learn about array formulas.

Count the occurrences of a condition

The COUNTIF worksheet function counts the occurrences of a value in a range of cells — for example, the number of cells in the range B5:B25 that contain the text "Northwind".

=COUNTIF(B5:B25,"Northwind")

Count the occurrences of multiple conditions

In the following formula, whenever Excel finds "Northwind" in the range B5:B25, it then checks for the text "Western" in the same row in column C (the range C5:C25). Excel then calculates the number of rows that contain both.

=SUM(IF(B5:B25="Northwind",IF(C5:C25="Western",1,0)))

This is an array formula and must be entered by pressing CTRL+SHIFT+ENTER. Learn about array formulas.

